# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=clash-prelude
pkgname=haskell-clash-prelude
pkgver=1.6.4
pkgrel=52
pkgdesc="Clash: a functional hardware description language - Prelude library"
url="https://github.com/clash-lang/clash-prelude"
license=('BSD')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-arrows' 'haskell-constraints' 'haskell-data-binary-ieee754'
         'haskell-data-default-class' 'haskell-extra' 'haskell-ghc-typelits-extra'
         'haskell-ghc-typelits-knownnat' 'haskell-ghc-typelits-natnormalise' 'haskell-half'
         'haskell-hashable' 'haskell-interpolate' 'haskell-lens' 'haskell-quickcheck'
         'haskell-recursion-schemes' 'haskell-reflection' 'haskell-singletons'
         'haskell-th-abstraction' 'haskell-th-lift' 'haskell-th-orphans' 'haskell-type-errors'
         'haskell-uniplate' 'haskell-vector')
makedepends=('ghc' 'uusi' 'haskell-doctest-parallel' 'haskell-hedgehog' 'haskell-hint'
             'haskell-quickcheck-classes-base' 'haskell-tasty' 'haskell-tasty-hedgehog'
             'haskell-tasty-hunit' 'haskell-tasty-quickcheck' 'haskell-tasty-th')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('6ed252593b11baa423b2068bd7704c2ac79006a286b7fea49d8a7ea42e50ef1748869b5987651277c31a7b6303f10ec9bbd4fe60d2e5f27c7fc8fd5ba9f0b580')

prepare() {
    cd $_hkgname-$pkgver
    gen-setup
    uusi -u lens $_hkgname.cabal
}

build() {
    cd $_hkgname-$pkgver
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    # TODO
    runhaskell Setup test --show-details=direct || warning "Tests failed"
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
